
algol,_n<
_b_e_g_i_n
   _c_o_m_m_e_n_t

        no buffer N=20:
        Time classic:   14.112
        Time turbo:     14.004 0.8pct

        buffer N=20:
        Time classic:   12.920
        Time turbo:     12.783 1.1pct

        buffer N=60:
        Time classic:   262.368
        Time turbo:     261.448 0.4pct
   ;
   _i_n_t_e_g_e_r _p_r_o_c_e_d_u_r_e LEQ1(N, M, A, eps);
   _v_a_l_u_e N, M, eps;
   _i_n_t_e_g_e_r N, M;
   _a_r_r_a_y A;
   _r_e_a_l eps;
   _b_e_g_i_n
      _i_n_t_e_g_e_r i, j, k, i1, j1;
      _r_e_a_l max, f2, factor;
      _i_n_t_e_g_e_r _a_r_r_a_y p[1:N];
      M := N + M;
      LEQ1 := 0;
      _f_o_r i := 1 _s_t_e_p 1 _u_n_t_i_l N _d_o
      _b_e_g_i_n
         max := 0;
         _f_o_r j := 1 _s_t_e_p 1 _u_n_t_i_l N _d_o
         max := max + A[i,j]|∧ 2;
         _i_f max > 1 ∨  max < 0.25 _t_h_e_n
         _b_e_g_i_n
            f2 := 2|∧ (−entier(ln(max)/1.3863 + 1));
            _f_o_r j := 1 _s_t_e_p 1 _u_n_t_i_l M _d_o
            A[i,j] := A[i,j]×f2
         _e_n_d if max
      _e_n_d for i: equilibration;
      _f_o_r k := 1 _s_t_e_p 1 _u_n_t_i_l N _d_o
      _b_e_g_i_n
         max := 0;
         _f_o_r i := k _s_t_e_p 1 _u_n_t_i_l N _d_o
         _f_o_r j := k _s_t_e_p 1 _u_n_t_i_l N _d_o
         _b_e_g_i_n
            factor := abs(A[i,j]);
            _i_f max _< factor _t_h_e_n
            _b_e_g_i_n
               max := factor;
               i1 := i;
               j1 := j
            _e_n_d if larger
         _e_n_d _f_o_r;
         _i_f max < eps _t_h_e_n
         _b_e_g_i_n
            LEQ1 := 1;
            _g_o_ _t_o EX
         _e_n_d error exit;
         max := A[i1,j1];
         _i_f i1 |= k _t_h_e_n
         _f_o_r j := k _s_t_e_p 1 _u_n_t_i_l M _d_o
         _b_e_g_i_n
            factor := A[k,j];
            A[k,j] := A[i1,j];



            A[i1,j] := factor
         _e_n_d for j: row interchange;
         p[k] := k;
         _i_f j1 |= k _t_h_e_n
         _b_e_g_i_n
            p[k] := j1;
            _f_o_r i := 1 _s_t_e_p 1 _u_n_t_i_l N _d_o
            _b_e_g_i_n
               factor := A[i,k];
               A[i,k] := A[i,j1];
               A[i,j1] := factor
            _e_n_d for i
         _e_n_d interchange of columns;
         _f_o_r i := k + 1 _s_t_e_p 1 _u_n_t_i_l N _d_o
         _b_e_g_i_n
            factor := A[i,k]/max;
            _f_o_r j := k + 1 _s_t_e_p 1 _u_n_t_i_l M _d_o
            A[i,j] := A[i,j] − A[k,j]×factor
         _e_n_d _f_o_r
      _e_n_d for k;
      _f_o_r k := N + 1 _s_t_e_p 1 _u_n_t_i_l M _d_o
      _f_o_r i := N _s_t_e_p −1 _u_n_t_i_l 1 _d_o
      _b_e_g_i_n
         factor := A[i,k];
         _f_o_r j := i + 1 _s_t_e_p 1 _u_n_t_i_l N _d_o
         factor := factor − A[i,j]×A[j,k];
         A[i,k] := factor/A[i,i]
      _e_n_d solving;
      _i_f M |= N _t_h_e_n
      _f_o_r i := N − 1 _s_t_e_p −1 _u_n_t_i_l 1 _d_o
      _b_e_g_i_n
         i1 := p[i];
         _i_f i1 |= i _t_h_e_n
         _f_o_r k := N + 1 _s_t_e_p 1 _u_n_t_i_l M _d_o
         _b_e_g_i_n
            factor := A[i,k];
            A[i,k] := A[i1,k];
            A[i1,k] := factor
         _e_n_d for k
      _e_n_d for i and solution interchange;
EX:_e_n_d LEQ1;
   _r_e_a_l _p_r_o_c_e_d_u_r_e clock count;
   _c_o_d_e clock count;
   1, 37;
     zl        , grf p−1   ; RF:=clock count; stack[p−1]:=RF;
   _e;
   _r_e_a_l _p_r_o_c_e_d_u_r_e RANDOM;
   _b_e_g_i_n
      _i_n_t_e_g_e_r new,mod;
      mod := 2796203;
      new := 125×oldrand;
      oldrand := new − mod×entier(new/mod);
      RANDOM := oldrand/mod−0.5
   _e_n_d RANDOM;
   _i_n_t_e_g_e_r oldrand,N;
   _r_e_a_l time,maxerror;

   select(17);
   oldrand:=100001;
   writecr;
   writetext(|<<N: |>);
   N:=60;
   writeinteger(|<p|>,N);
   _b_e_g_i_n



      _a_r_r_a_y A[1:N,1:N+1];
      _i_n_t_e_g_e_r i,j;
      _r_e_a_l sum;
      _f_o_r i:=1 _s_t_e_p 1 _u_n_t_i_l N _d_o
      _b_e_g_i_n
         sum:=0;
         _f_o_r j:=1 _s_t_e_p 1 _u_n_t_i_l N _d_o
         _b_e_g_i_n
            A[i,j]:=RANDOM;
            sum:=sum+A[i,j]
         _e_n_d;
         A[i,N+1]:=sum
      _e_n_d;
      writecr;
      clock count;
      _i_f LEQ1(N, 1, A, 110−12)=1 _t_h_e_n
      writetext(|<<Error.|>);
      time:=clock count;
      write(|<ddddd.ddd|>,time);
      writecr;
      maxerror:=0;
      _f_o_r i:=1 _s_t_e_p 1 _u_n_t_i_l N _d_o
      _b_e_g_i_n
         sum:=abs(A[i,N+1]−1);
         _i_f sum>maxerror _t_h_e_n maxerror:=sum
      _e_n_d _f_o_r;
      write(|<d.ddddd10−dd|>,maxerror)
   _e_n_d inner block
_e_n_d
t<


